

Supplementary Material to “A likelihood-based two-part marginal model for longitudinal semicontinuous data”

Li Su*, Brian D. M. Tom and Vernon T. Farewell

Medical Research Council Biostatistics Unit, Robinson Way, Cambridge CB2 0SR, UK

1 SAS Program for the HAQ Analysis

```
title 'Marginally specified two-part model for the HAQ data';
```

```
PROC NLMIXED DATA=haq.haq_final;
```

```
*INITIAL VALUES;
```

```
PARMS  theta0=0.60,theta1=0.50,theta2=-0.20, theta3=-0.50,  
theta4=0.80,theta5=0.40,theta6=0.20, theta7=1.20,  
beta0=0.50, beta1=0.20, beta2=0.10, beta3=0.00, beta4=0.00,  
beta5=0.10,beta6=0.05, beta7=0.35,  
sigmae=0.08,sigmav=0.25,logsigmab=2.25, r=0.97;
```

```
*SET BOUNDS FOR VARIANCE COMPONENTS
```

```
BOUNDS sigmae sigmav > 0,-1<=r<=1;
```

```
*BRIDGE DISTRIBUTION FOR RANDOM INTERCEPT IN BINARY PART;
```

```
sigmab=exp(logsigmab);
```

```
uni=probnorm(U);
```

```
pi=4*ATAN(1);
```

```
phi=1/sqrt(1+3*sigmab/(pi*pi));
```

*Address for correspondence: Li Su, MRC Biostatistics Unit, Robinson Way, Cambridge CB2 0SR, UK.
Email: li.su@mrc-bsu.cam.ac.uk; phone: 44-1223-760722; fax: 44-1223-760729

```

Bl=1/phi*log(sin(pi*uni*phi)/sin(phi*pi*(1-uni)));

* LOGIT LINK FOR BINARY PART;

LOGITP=(theta0+theta1*B27+theta2*DQW3+theta3*DR7+theta4*DQW3*DR7+
theta5*AGEPSA2+theta6*ARTHDUR3+theta7*SEX)/phi+B1;

* PROBABILITY OF NON-ZERO HAQ;

P=1/(1+exp(-LOGITP));

* IDENTITY LINK FOR CONTINUOUS PART;

MU=beta0+beta1*B27+beta2*DQW3+beta3*DR7+beta4*DQW3*DR7+
beta5*AGEPSA2+beta6*ARTHDUR3+beta7*SEX*DEFORMED2+V;

*LOG LIKELIHOOD;

LOGLIKE=(1-HAQBIN)*log(1-P)+(HAQBIN)*(log(P)-0.5*((HAQ-MU)/sqrt(sigmae))**2
- log(sqrt(sigmae)*SQRT(2*pi)));

MODEL HAQ ~ GENERAL(LOGLIKE);

RANDOM U V ~ NORMAL([0,0],[1,r*(sigmav)**0.50,sigmav]) SUBJECT=PTNO;

* ESTIMATE ATTENUATION FACTOR;

ESTIMATE 'phi' phi;

* VARIANCE FOR BRIDGE DISTRIBUTION;

ESTIMATE 'sigb2' exp(logsigmab);

* SUBJECT-SPECIFIC COVARIATE EFFECTS;

ESTIMATE 'intercept' theta0/phi;

ESTIMATE 'b27' theta1/phi;

ESTIMATE 'dqw3' theta2/phi;

```

```
ESTIMATE 'dr7' theta3/phi;  
ESTIMATE 'dqw3:dr7' theta4/phi;  
ESTIMATE 'Onset age of PSA' theta5/phi;  
ESTIMATE 'Arthritis duration' theta6/phi;  
ESTIMATE 'Sex' theta7/phi;  
RUN;
```